;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           #\# #\> (lambda (s c p) (declare (ignore c p)) (read s nil nil t))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(declaim (inline find-argopt))
(defun find-argopt (sequence predicate &key (start 0) end (key #'identity))
  "Returns an index x at which SEQUENCE takes the minimal (or maximal, depending
on PREDICATE) value, and returns SEQUENCE[x] as the second value.

To explain the behaviour briefly, when (FUNCALL PREDICATE (AREF SEQUENCE
<index>) (AREF SEQUENCE <index of current optimum>)) holds, <index> and <index
of current optimum> are swapped."
  (declare ((or null (integer 0 #.most-positive-fixnum)) end)
           ((integer 0 #.most-positive-fixnum) start)
           (sequence sequence))
  (labels ((invalid-range-error ()
             (error "Can't find optimal value in null interval [~A, ~A) on ~A" start end sequence)))
    (etypecase sequence
      (list
       (let ((sequence (nthcdr start sequence))
             (end (or end most-positive-fixnum)))
         (when (or (null sequence)
                   (>= start end))
           (invalid-range-error))
         (let ((opt-element (car sequence))
               (opt-index start)
               (pos start))
           (dolist (x sequence)
             (when (>= pos end)
               (return-from find-argopt (values opt-index opt-element)))
             (when (funcall predicate (funcall key x) (funcall key opt-element))
               (setq opt-element x
                     opt-index pos))
             (incf pos))
           (values opt-index opt-element))))
      (vector
       (let ((end (or end (length sequence))))
         (when (or (>= start end)
                   (>= start (length sequence)))
           (invalid-range-error))
         (let ((opt-element (aref sequence start))
               (opt-index start))
           (loop for i from start below end
                 for x = (aref sequence i)
                 do (when (funcall predicate (funcall key x) (funcall key opt-element))
                      (setq opt-element x
                            opt-index i)))
           (values opt-index opt-element)))))))

(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  (declare #.OPT)
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10)) result))))
              (return (if minus (- result) result))))))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(define-modify-macro minf (new-value) min)

(defun main ()
  (declare #.OPT)
  (let* ((n (read))
         (m (read))
         (graph (make-array n :element-type 'list :initial-element nil))
         (out-degrees (make-array n :element-type 'uint31))
         (base-exps (make-array n :element-type 'double-float :initial-element 0d0)))
    (declare ((simple-array list (*)) graph)
             (uint31 n m))
    (dotimes (i m)
      (let ((src (- (read-fixnum) 1))
            (dest (- (read-fixnum) 1)))
        (push dest (aref graph src))
        (incf (aref out-degrees src))))
    (loop for v from (- n 2) downto 0
          do (let ((value 0d0))
               (declare (double-float value))
               (dolist (next (aref graph v))
                 (incf value (aref base-exps next)))
               (setf (aref base-exps v)
                     (+ 1d0 (/ value (aref out-degrees v))))))
    (let ((res (aref base-exps 0))
          (exps (make-array n :element-type 'double-float)))
      (declare (double-float res)
               ((simple-array double-float (*)) exps))
      (loop
        for pivot from (- n 2) downto 0
        unless (null (cdr (aref graph pivot)))
        do (fill exps 0d0)
           (loop
             for v from (- n 2) downto 0
             for odeg = (aref out-degrees v)
             do (if (= pivot v)
                    (let ((value 0d0)
                          (excluded (nth-value 1
                                               (find-argopt (aref graph v)
                                                            #'>
                                                            :key (lambda (next)
                                                                   (aref base-exps next))))))
                      (declare (double-float value)
                               (uint31 excluded))
                      (dolist (next (aref graph v))
                        (declare (uint31 next))
                        (unless (= next excluded)
                          (incf value (aref exps next))))
                      (setf (aref exps v)
                            (+ 1d0 (/ value (- odeg 1)))))
                    (let ((value 0d0))
                      (declare (double-float value))
                      (dolist (next (aref graph v))
                        (incf value (aref exps next)))
                      (setf (aref exps v)
                            (+ 1d0 (/ value odeg))))))
           (minf res (aref exps 0)))
      (println res))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "C:/msys64/usr/bin/cat.exe" '("/dev/clipboard") :output out)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "4 6
1 4
2 3
1 3
1 2
3 4
2 4
"
    "1.5000000000
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "3 2
1 2
2 3
"
    "2.0000000000
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "10 33
3 7
5 10
8 9
1 10
4 6
2 5
1 7
6 10
1 4
1 3
8 10
1 5
2 6
6 9
5 6
5 8
3 6
4 8
2 7
2 9
6 7
1 2
5 9
6 8
9 10
3 9
7 8
4 5
2 10
5 7
3 5
4 7
4 9
"
    "3.0133333333
")))


Measured memory_bytes: 20968